Übungen zu Organisation und Architektur von Rechnern SS 2002

Serie 5 — Musterlösung

Aufgabe 14 (10 Punkte)

Die in der Aufgabenstellung geforderte Aufrufkonvention für die Funktion `FAC` wird wie folgt realisiert:

Der aufrufende Kontext übergibt den Parameter über den Stack. Bei Eintritt in die Funktion `FAC` liegt auf der Spitze des Stack also die Rücksprungadresse und darunter das Argument. Um im weiteren Verlauf elegant auf das Argumentframe zugreifen zu können, wird mit Hilfe der `LINK` Instruktion die initiale Adresse des Stackpointers `SP` im Register `A6` gesichert. Dadurch kann unabhängig von der späteren Position des `SP` immer mittels `8(A6)` auf das Argument zugegriffen werden. Als Seiteneffekt sichert `LINK` den aktuellen Wert von `A6` auf dem Stack, damit er beim Verlassen der Funktion wiederhergestellt werden kann.

Um die Anforderung, alle Register nach dem Funktionsaufruf unverändert zu hinterlassen, erfüllen zu können, werden anschließend unter Verwendung der Anweisung `MOVEM` alle durch die Funktion verwendeten Register auf dem Stack gesichert.

Am Ende der Funktion wird der Rückgabewert in das Argumentframe geschrieben und es werden die Register wieder mit Hilfe von `MOVEM` in ihren ursprünglichen Zustand zurückversetzt. Der abschließende `UNLK` Befehl wirkt komplementär zum `LINK` und holt den alten Wert des Registers `A6` vom Stack.

Nach dem Rücksprung mittels `RTS` befindet sich also, wie in der Aufgabenstellung gefordert, nur noch das Ergebnis der Berechnung oben auf dem Stack.

Nach selbigem Schema muß die Funktion `MULU_L` modifiziert werden. Hier der vollständige Quelltext:

```
1                 ORG     $0
2                 DC.L    $8000           Stack pointer value after a reset
3                 DC.L    START           Program counter value after a reset
4
5   START:        MOVE.L  #12,-(SP)       push the argument on the stack
6                 JSR     FAC             call factorial
7                 MOVE.L  (SP)+,D0        pop the result from the stack
8                 BREAK
9
10                MOVE.L  #13,-(SP)
11                JSR     FAC
12                MOVE.L  (SP)+,D0
13                BREAK
14
15
16  ********************************************************************************
17  *
18  * function:   (SP) FAC( (SP))
19  *
20  * description:
21  *   computes the factorial of the number that was top of the runtime stack
22  *   right before calling the function. The result is returned via the very
23  *   same stack location.
24  *
25  * registers affected: none
26  *
27
28  FAC:            LINK    A6,#0           save A6 and set it to its stack location
29                  MOVEM.L D0-D1,-(SP)     save the registers used
30                  MOVE.L  8(A6),D0        fetch n
31
32                  CMPI    #0,D0           n == 0?
33                  BNE     ELSE
34                  MOVE.L  #1,D0           res (D0) = 1
35                  BRA     DONE
36  ELSE:           MOVE.L  D0,D1
37                  SUBI    #1,D1
```

```
38                  MOVE.L  D1,-(SP)        push (n-1)
39                  JSR     FAC             call factorial
40                  MOVE.L  D0,-(SP)
41                  JSR     MULU_L
42                  MOVE.L  (SP)+,D0        higher long of the result
43                  BEQ     OK
44                  MOVE.L  #0,D0           signalize an error
45                  BRA     DONE
46  OK:             MOVE.L  (SP)+,D0        lower long of the result
47
48  DONE:           MOVE.L  D0,8(A6)        save the result in the argument frame
49                  MOVEM.L (SP)+,D0-D1     restore the registers used
50                  UNLK    A6              restore A6
51                  RTS
52
53
54  ********************************************************************************
55  *
56  * function:   (SP1), (SP2) MULU_L( (SP1), (SP2))
57  *
58  * description:
59  *   multiplies two long formats given in (SP1),(SP2). The result is returned
60  *   in (SP2),(SP1), which hold the higher and the lower long format of the
61  *   result, respectively.
62  *
63  * registers affected: none
64  *
65
66  MULU_L:         LINK    A6,#0
67                  MOVEM.L D0-D7,-(SP)
68                  MOVE.L  12(A6),D0
69                  MOVE.L  8(A6),D1
70
71                  MOVE.L  D0,D4
72                  SWAP    D4              D4 = HW( D0)
73                  MOVE.L  D1,D5
74                  SWAP    D5              D5 = HW( D1)
75                  MOVE.L  D4,D2
76                  MULU    D5,D2           D2 = HW( D0) * HW( D1)
77                  MOVE.L  D1,D3
78                  MULU    D0,D3           D3 = LW( D0) * LW( D1)
79
80                  MULU    D1,D4           D4 = HW( D0) * LW( D1)
81                  MULU    D0,D5           D5 = LW( D0) * HW( D1)
82                  ADD.L   D5,D4
83                  BCC     NO_INC_HW       do we have a carry affecting r0?
84                  ADD.L   #$10000,D2      increment r0
85  NO_INC_HW:      SWAP    D4
86                  MOVE.L  D4,D5
87                  ANDI.L  #$FFFF,D4       D4 = $0000     HW(D4+D5)
88                  ANDI.L  #$FFFF0000,D5   D5 = LW(D4+D5) $0000
89                  ADD.L   D5,D3           LL( D0.L * D1.L)
90                  ADDX.L  D4,D2           HL( D0.L * D1.L)
91
92                  MOVE.L  D2,8(A6)        store D2
93                  MOVE.L  D3,12(A6)       store D3
94                  MOVEM.L (SP)+,D0-D7
95                  UNLK    A6
96                  RTS
```

Aufgabe 15 (10 Punkte)

Für die Funktionen `F` und `G` des Roller-Coasters wird das gleiche Programmgerüst wie für `FAC` in Aufgabe 14 verwendet. Da der `LINK` Befehl den aktuellen Wert des Registers `A6` auf dem Stack speichert, entsteht als Seiteneffekt von `LINK` eine Verkettung der Argumentframes. Dazu muß vor dem Aufruf einer Funktion das Register `A6` auf das zu der aufgerufenen Funktion übergeordnete Argumentframe zeigen. Folgende Fälle sind dabei zu unterscheiden:

a) *Aufruf einer Funktion, die lokal zur aktuellen Funktion definiert ist:*

In diesem Fall zeigt `A6` bereits auf das übergeordnete Frame, da die aktuelle Funktion auch die übergeordnete ist.

Dies tritt im vorliegenden Beispiel auf, wenn die Funktion `G` innerhalb der Funktion `F` aufgerufen wird (Zeilen 36 f. im Quellcode).

b) *Aufruf einer Funktion, die auf gleicher Ebene definiert ist:*

Hier ist die der aktuellen Funktion übergeordnete Funktion auch die übergeordnete Funktion der aufzurufenden Funktion. Daher muß `A6` einmal dereferenziert werden, um auf das übergeordnete Frame zu zeigen.

Dieser Fall tritt im vorliegenden Beispiel ein, wenn die Funktion `G` sich selbst aufruft (Zeilen 91 f. im Quellcode).

c) *Aufruf einer Funktion, zu der die aktuelle Funktion lokal definiert ist:*

Da die aufgerufene Funktion der aufrufenden übergeordnet ist, muß `A6` mehrfach dereferenziert werden, abhängig davon, wie viele Ebenen die aufgerufene Funktion über der aufrufenden definiert wurde.

Beim Roller-Coaster ist dies dann der Fall, wenn die Funktion `G` die Funktion `F` aufruft. `F` ist eine Ebene über `G` definiert und somit muß `A6` zweimal dereferenziert werden (Zeilen 79 ff. im Quellcode).

Diese Regeln sind vollkommen analog zu den in der Vorlesung vorgestellten Regeln für die `SECDH-2` Maschine.

In den Fällen b) und c) ist sicherzustellen, daß die aufrufende Funktion den Wert von `A6` (auf dem Stack) sichert, bevor sie ihn für die aufzurufende Funktion modifiziert (z. B. Zeile 75), und ihn nach der Rückkehr aus der Unterfunktion wiederherstellt (z. B. Zeile 83). Ansonsten hätte sie anschließend zum einen keine Möglichkeit mehr, über `A6` auf ihr eigenes Argumentframe zuzugreifen, und zum anderen würde `UNLK` den `SP` auf einen falschen Wert setzen.

Um Zugriffe auf relativ freie Variablen zu realisieren, muß die durch `A6` referenzierte verkettet Liste von Argumentframes durchlaufen werden, bis die jeweilige Bindungsebene der Variablen erreicht ist (vgl. Zeilen 63 ff. im Quellcode).

Beim Aufruf von `F` mit den Argumenten 1, 2, 4 entsteht folgende Struktur auf dem Laufzeitstack:



Hier der Quellcode des Roller-Coasters:

```
1                ORG     $0
2                DC.L    $8000           stack pointer value after a reset
3                DC.L    START           program counter value after a reset
4
5   START:       MOVE.L  #1,-(SP)        put 1 into arg. frame for F
6                MOVE.L  #2,-(SP)        put 2 into arg. frame for F
7                MOVE.L  #20,-(SP)       put 20 into arg. frame for F
8                LEA.L   $0,A6           initialize frame pointer
9                                        * (there is no higher arg. frame yet)
```

```
10                JSR     F               call F( 1, 2, 20)
11                MOVE.L  (SP)+,D0        put result into D0
12                BREAK
13
14
15  ********************************************************************************
16  *
17  * function: (SP1) F( (SP1), (SP2), (SP3))
18  *
19  * description:
20  *   Expects an argument frame with three arguments on the stack.
21  *   The register A6 should contain the frame pointer of the actual environment.
22  *
23  * registers affected: none
24  *
25
26  F:              LINK    A6,#0           store pointer to arg. frame in A6
27                  MOVEM.L D0-D2,-(SP)     save registers
28                  MOVE.L  16(A6),D0       fetch u from actual arg. frame
29                  MOVE.L  12(A6),D1       fetch v from actual arg. frame
30                  MOVE.L  8(A6),D2        fetch n from actual arg. frame
31
32                  SUBI.L  #1,D2           n = n - 1
33                  MOVE.L  D1,-(SP)        put v into arg. frame for G
34                  MOVE.L  D0,-(SP)        put u into arg. frame for G
35                  MOVE.L  D2,-(SP)        put n into arg. frame for G
36                                          * no frame pointer adjustment needed
37                  JSR     G               call G( v, u, n)
38                  MOVE.L  (SP)+,D0        put result into D0
39                  MOVE.L  D0,16(A6)       write result into arg. frame (1st entry)
40
41                  MOVEM.L (SP)+,D0-D2     restore registers
42                  UNLK    A6              restore A6
43                  MOVE.L  (SP),8(SP)      kill last two entries of arg. frame:
44                  ADDQ.L  #8,SP             move return address, adjust SP
45                  RTS
46
47
48  ********************************************************************************
49  *
50  * function: (SP1) G( (SP1), (SP2), (SP3))
51  *
52  * description:
53  *   Expects an argument frame with three arguments on the stack.
54  *   The register A6 should contain the frame pointer of the actual environment.
55  *
56  * registers affected: none
57  *
58
59  G:              LINK    A6,#0           store pointer to arg. frame in A6
60                  MOVEM.L D0-D4,-(SP)     save registers
61                  MOVE.L  16(A6),D0       fetch w from actual arg. frame
62                  MOVE.L  12(A6),D1       fetch z from actual arg. frame
63                  MOVE.L  8(A6),D2        fetch n from actual arg. frame
64                  MOVEA.L (A6),A0         fetch pointer to arg. frame of F
65                  MOVE.L  16(A0),D3       fetch u from arg. frame of F
66                  MOVE.L  12(A0),D4       fetch v from arg. frame of F
67
68                  CMPI.L  #0,D2           (n == 0) ?
69                  BLE     G_TAIL
70                  SUBI.L  #1,D2           n = n - 1
71                  CMP.L   D3,D0           (w > u) ?
72                  BGT     G_G
73
74  G_F:            ADDI.L  #1,D0           w = w + 1
75                  MOVE.L  A6,-(SP)        save A6
76                  MOVE.L  D4,-(SP)        put v into arg. frame for F
77                  MOVE.L  D0,-(SP)        put w into arg. frame for F
78                  MOVE.L  D2,-(SP)        put n into arg. frame for F
79                  MOVEA.L (A6),A6         adjust frame pointer for F
80                  MOVEA.L (A6),A6         adjust frame pointer for F
81                  JSR     F               call F( v, w, n)
82                  MOVE.L  (SP)+,D0        res = F( ...)
83                  MOVE.L  (SP)+,A6        restore A6
```

```
                JMP     G_END

G_G:            SUBI.L  #1,D3           u = u - 1
                MOVE.L  A6,-(SP)        save A6
                MOVE.L  D3,-(SP)        put u into arg. frame for G
                MOVE.L  D1,-(SP)        put z into arg. frame for G
                MOVE.L  D2,-(SP)        put n into arg. frame for G
                MOVEA.L (A6),A6         adjust frame pointer for G
                JSR     G               call G( u, z, n)
                MOVE.L  (SP)+,D0        res = G( ...)
                MOVE.L  (SP)+,A6        restore A6
                JMP     G_END

G_TAIL:         ADD.L   D1,D0           res = w + z

G_END:          MOVE.L  D0,16(A6)       write result into arg. frame (1st entry)
                MOVEM.L (SP)+,D0-D4     restore registers
                UNLK    A6              restore A6
                MOVE.L  (SP),8(SP)      kill last two entries of arg. frame:
                ADDQ.L  #8,SP             move return address, adjust SP
                RTS


*****************************************************************************


END
```